KKS2025, Conjecture 23, eq. (10.9)

In[ ]:=
```
<< RISC`HolonomicFunctions`;
<< RISC`Guess`;
```

HolonomicFunctions Package version 1.7.3 (21-Mar-2017)
written by Christoph Koutschan
Copyright Research Institute for Symbolic Computation (RISC),
Johannes Kepler University, Linz, Austria

```
--> Type  ?HolonomicFunctions  for help.
```

Package GeneratingFunctions version 0.9 written by Christian Mallinger
Copyright Research Institute for Symbolic Computation (RISC),
Johannes Kepler University, Linz, Austria

Guess Package version 0.52
written by Manuel Kauers
Copyright Research Institute for Symbolic Computation (RISC),
Johannes Kepler University, Linz, Austria

In[ ]:=
```
SetDirectory[NotebookDirectory[]];
```

The following initializing codes are taken from Christoph Koutschan, Christian Krattenthaler and Michael Schlosser's implementation for their 2025 JSC paper on determinant evaluations.
http://www.koutschan.de/data/det3/

Reference

C. Koutschan, C. Krattenthaler, and M. J. Schlosser, Determinant evaluations inspired by Di Francesco's determinant for twenty-vertex configurations, *J. Symbolic Comput.* **127** (2025), Paper No. 102352, 34 pp.
https://doi.org/10.1016/j.jsc.2024.102352

In[ ]:=
```
(* Display all relevant information about an annihilator ideal. *)
AnnInfo[ann_] := With[{vars = First /@ OreAlgebra[ann][[1]]}, Print[
    "ByteCount: ", ByteCount[ann],
    "\nSupport: ", Support[ann],
    "\ndegree " <> ToString[vars] <> ": ", Exponent[#, vars] & /@ ann,
    "\nStandard Monomials: ", UnderTheStaircase[ann],
    "\nHolonomic Rank: ", Length[UnderTheStaircase[ann]]
   ]];
```

```
In[•]:= (* A straight-
   forward implementation of reduction modulo a left ideal in the shift algebra. *)
  (* Reason: the built-in procedure "OreReduce"
     in the HolonomicFunctions package sometimes
     causes Mathematica to crash. *)
  SortLex[m1_, m2_] := With[{f1 = First[m1], f2 = First[m2]},
     If[f1 =!= f2 || Length[m1] === 1, f1 > f2, SortLex[Rest[m1], Rest[m2]]]];
  SortDLex[m1_, m2_] := With[{w1 = Plus @@ m1, w2 = Plus @@ m2},
     If[w1 === w2, SortLex[m1, m2], w1 > w2]];
  Add[p1_List, p2_List] :=
    Module[{p = {}, c, i1 = 1, i2 = 1, l1 = Length[p1], l2 = Length[p2], e1, e2},
     While[i1 ≤ l1 && i2 ≤ l2,
      {e1, e2} = {p1[[i1, 2]], p2[[i2, 2]]};
      Which[
       e1 === e2, If[(c = p1[[i1, 1]] + p2[[i2, 1]]) =!= 0, AppendTo[p, {c, e1}]];
       i1++; i2++;
       ,
       SortDLex[e1, e2], AppendTo[p, p1[[i1]]]; i1++;
       ,
       SortDLex[e2, e1], AppendTo[p, p2[[i2]]]; i2++;
      ];
     ];
     If[i1 ≤ l1, p = Join[p, Take[p1, {i1, l1}]]];
     If[i2 ≤ l2, p = Join[p, Take[p2, {i2, l2}]]];
     Return[p];
    ];
  ScalarMult[s_, p_List] := {Expand[Together[s * #1]], #2} & @@@ p;
  OreReduce1[p_List, g_List] := OreReduce1[#, g] & /@ p;
  OreReduce1[p1_OrePolynomial, g1 : {(_OrePolynomial) ..}] :=
    Module[{p = p1, g = g1, v, e, f, f1, r = {}, k, gk, gcd},
     v = First /@ OreAlgebra[p][[1]];
     {p, g} = {First[p], First /@ g};
     f = PolynomialLCM @@ (Denominator[First[#]] & /@ p);
     p = ScalarMult[f, p];
     While[p =!= {},
      k = 1;
      While[Min[e = (p[[1, 2]] - g[[k, 1, 2]])] < 0, k++];
      If[k > Length[g],
       AppendTo[r, p[[1]]];
       p = Rest[p];
       ,
       gk = {Expand[#1 /. Thread[v → (v + e)]], #2 + e} & @@@ g[[k]];
       gcd = PolynomialGCD[p[[1, 1]], gk[[1, 1]]];
       f *= (f1 = Together[gk[[1, 1]] / gcd]);
       gk = ScalarMult[Together[-p[[1, 1]] / gcd], Rest[gk]];
       p = Add[ScalarMult[f1, Rest[p]], gk];
      ];
     ];
     Return[OrePolynomial[{Together[#1 / f], #2} & @@@ r, p1[[2]], p1[[3]]]];
    ];
```

```
In[∘]:= ClearAll[prod];

prodsimp = {prod[a_, {i_, b_}] → prod[a, {i, 1, b}],
   prod[a_, {i_, b0_, b1_}] / prod[a_, {i_, b0_, b2_}] /; IntegerQ[Expand[b1 - b2]] :→
    If[Expand[b1 - b2] ≥ 0, Product[a, {i, b2 + 1, b1}], 1 / Product[a, {i, b1 + 1, b2}]],
   prod[a1_, b_]^e1_. * prod[a2_, b_]^e2_. :→ prod[FunctionExpand[a1^e1 * a2^e2], b]};
```

Initialization

Set up the determinant (of matrix $a_{i,j}$) in question.

```
In[∘]:= ClearAll[mata, mata1, mata2, matc, datac, prodform];

In[∘]:= ClearAll[a, b, c, d, e, f, i, j, n];

Print["We are going to evaluate the determinant:\n",
  TraditionalForm[HoldForm @@ {Subscript[det, 0 ≤ i, j < n][
      e^ (i + b) Binomial[f * j + i + c, f * j + a] + Binomial[f * j - i + d, f * j + a]]}], "\n"];

{a, b, c, d} = {2, 1, 2, 0};
{e, f} = {2, 4};

mata1[i_, j_] := e^ (i + b) Binomial[f * j + i + c, f * j + a];
mata2[i_, j_] := Binomial[f * j - i + d, f * j + a];
mata[i_, j_] := mata1[i, j] + mata2[i, j];
mata[i_Integer, j_Integer] := FunctionExpand[mata1[i, j] + mata2[i, j]];

prodform[0] = 1;
SetDelayed @@
 (Hold[prodform[n_], If[IntegerQ[n], FunctionExpand[C /. prod → Product], C]] /.

   {C → Gamma[n + 1] / Gamma[2 n + 1] * prod[(Gamma[6 i - 1] Gamma[(i+2)/4]) / (Gamma[5 i - 1] Gamma[(5 i-2)/4]), {i, 1, n}]});

Print[">>> With the following choice of parameters:\n",
  "{a, b, c, d} = ", {a, b, c, d}, ";\n", "{l, m} = ",
  {e, f}, ";\n\nWe are going to prove:\n", TraditionalForm[
   HoldForm @@ {Subscript[det, 0 ≤ i, j < n][e^ (i + b) Binomial[f * j + i + c, f * j + a] +
        Binomial[f * j - i + d, f * j + a]] == prodform[n] /. prod → Product}], "\n"];

Print["The matrix of ", Subscript["a", "i,j"], " begins with:\n",
  TableForm[Table[mata[i, j], {i, 0, 5}, {j, 0, 5}]], "\n"];

Print["The determinants begin with:\n",
  Table[Det[Table[mata[i, j], {i, 0, n - 1}, {j, 0, n - 1}]], {n, 1, 6}], "\n"];

Print["The product formula begins with:\n", Table[prodform[n], {n, 1, 6}]];
```

We are going to evaluate the determinant:

$$\det_{0\le i,j<n}\left(\binom{d-i+fj}{a+fj}+e^{b+i}\binom{c+i+fj}{a+fj}\right)$$

```
>>> With the following choice of parameters:
{a, b, c, d} = {2, 1, 2, 0};
{l, m} = {2, 4};

We are going to prove:
```

$$\det_{0\le i,j<n}\left(\binom{-i+4j}{2+4j}+2^{1+i}\binom{2+i+4j}{2+4j}\right)=\frac{\Gamma(1+n)\prod_{i=1}^{n}\frac{\Gamma\left(\frac{2+i}{4}\right)\Gamma(-1+6i)}{\Gamma\left(\frac{1}{4}(-2+5i)\right)\Gamma(-1+5i)}}{\Gamma(1+2n)}$$

The matrix of $a_{i,j}$ begins with:

2	2	2	2	2	2
13	28	44	60	76	92
51	224	528	960	1520	2208
166	1344	4576	10 880	21 280	36 800
490	6720	32 032	97 920	234 080	478 400
1359	29 569	192 192	744 192	2 153 536	5 166 720

The determinants begin with:
{2, 30, 3584, 3 424 256, 26 172 456 960, 1 599 974 638 878 720}

The product formula begins with:
{2, 30, 3584, 3 424 256, 26 172 456 960, 1 599 974 638 878 720}

Construct the minor-related quantity $c_{n,j}$.

We will generate the data of $c_{n,j}$ in advance. **No need to execute the following codes again.** Instead, import the data directly.

In[∘]:=
```
start = CurrentDate[];

ClearAll[DATAC, MATC];

MAX = 70;

DATAC[n_Integer] := DATAC[n] =
   With[{ns = NullSpace[Table[mata[i, j], {i, 0, n - 2}, {j, 0, n - 1}]][[1]]},
    Together[ns / Last[ns]]];
MATC[n_Integer, j_Integer] := Which[n == 1 && j == 0, 1, j ≥ n, 0, True, DATAC[n][[j + 1]]];

Export["datac.txt", {Table[MATC[n, j], {n, MAX}, {j, 0, n - 1}]}]

Print["Time used: ", CurrentDate[] - start];
```

Out[∘]= datac.txt

Time used: 22.1316 s

Import the data of $c_{n,j}$.

In[∘]:=
```
DATACImported = ToExpression[Import["datac.txt"]];

datac[n_Integer] := datac[n] = DATACImported[[n]];
matc[n_, j_] := matc[n, j] = Piecewise[{{datac[n][[j + 1]], j < n}}, 0];

Print["The matrix of ", Subscript["c", "n,j"],
  " begins with:\n", TableForm[Table[matc[n, j], {n, 1, 6}, {j, 0, n - 1}]]];
```

The matrix of $c_{n,j}$ begins with:

1					
-1	1				
$\frac{16}{15}$	$-\frac{31}{15}$	1			
$-\frac{8}{7}$	$\frac{45}{14}$	$-\frac{43}{14}$	1		
$\frac{256}{209}$	$-\frac{929}{209}$	$\frac{1315}{209}$	$-\frac{851}{209}$	1	
$-\frac{256}{195}$	$\frac{1124}{195}$	$-\frac{419}{39}$	$\frac{2021}{195}$	$-\frac{989}{195}$	1

Guess the annihilator for $c_{n,j}$.

We will generate the guessed annihilator for $c_{n,j}$ in advance. **No need to execute the following codes again.** Instead, import the data directly.

```
In[•]:= start = CurrentDate[];

MAX = 60;

ClearAll[cc, n, j];

guess =
  GuessMultRE[Table[Piecewise[{{matc[n, j], j ≤ n - 1}}, 0], {n, 1, MAX}, {j, 0, MAX - 1}],
   Flatten[Table[cc[n + l1, j + l2], {l1, 0, 3}, {l2, 0, 4}]],
   {n, j}, 8, StartPoint → {1, 0}, Constraints → (j < n)];

Print["Time used: ", CurrentDate[] - start];
```

Time used: 2.00584 min

```
In[•]:= start = CurrentDate[];

annc = OreGroebnerBasis[NormalizeCoefficients /@ ToOrePolynomial[guess, cc[n, j]]];
AnnInfo[annc]
Export["annc.txt", {annc}]

Print["Time used: ", CurrentDate[] - start];
```

ByteCount: 264 912

Support: $\{\{S_n^2, S_n S_j, S_j^2, S_n, S_j, 1\}, \{S_j^3, S_n S_j, S_j^2, S_n, S_j, 1\}, \{S_n S_j^2, S_n S_j, S_j^2, S_n, S_j, 1\}\}$

degree {n, j}: {{16, 10}, {9, 9}, {7, 2}}

Standard Monomials: $\{1, S_j, S_n, S_j^2, S_n S_j\}$

Holonomic Rank: 5

Out[•]= annc.txt

Time used: 2.36354 min

Import the annihilator for $c_{n,j}$.

In[∘]:=
```
ClearAll[n, j, cc];

annc = ToExpression[Import["annc.txt"]];

AnnInfo[annc]

Print[];

MAX = 6;
Print["Check whether the first values of ",
  Subscript["c", "n,j"], " satisfy the guessed recurrences:\n",
  Union[Flatten[Table[Together[ApplyOreOperator[annc, cc[n, j]] /.
      {n → nn, j → jj, cc → matc}], {nn, 1, MAX}, {jj, 0, nn - 1}]]]];

Print[];

Print["The values at these indices have to be given as initial conditions,
    in order to uniquely define ",
  Subscript["c", "n,j"], " via the recurrences in annc:\n",
  AnnihilatorSingularities[annc, First /@ OreAlgebra[annc][[1]] /. {n → 1, j → 0}]];
```

ByteCount: 264 336

Support: $\{\{S_n^2, S_n S_j, S_j^2, S_n, S_j, 1\}, \{S_j^3, S_n S_j, S_j^2, S_n, S_j, 1\}, \{S_n S_j^2, S_n S_j, S_j^2, S_n, S_j, 1\}\}$

degree {n, j}: {{16, 10}, {9, 9}, {7, 2}}

Standard Monomials: $\{1, S_j, S_n, S_j^2, S_n S_j\}$

Holonomic Rank: 5

Check whether the first values of $c_{n,j}$ satisfy the guessed recurrences:

{0}

The values at these indices have to be given as initial conditions,
  in order to uniquely define $c_{n,j}$ via the recurrences in annc:

{{{j → 0, n → 1}, True}, {{j → 0, n → 2}, True},
  {{j → 1, n → 1}, True}, {{j → 1, n → 2}, True}, {{j → 2, n → 1}, True}}

Proof of (H1)

Compute a recurrence for $c_{n,n-1}$.

In[∘]:=
```
start = CurrentDate[];

ClearAll[n, j];
Support[cnn1 = DFiniteSubstitute[annc, {j → n - 1}][[1]]]

Print["Time used: ", CurrentDate[] - start];
```

Out[∘]= $\{S_n^5, S_n^4, S_n^3, S_n^2, S_n, 1\}$

Time used: 24.1019 s

Verify that this recurrence admits a constant sequence as solution.

In[∘]:=
```
OreReduce1[cnn1, Annihilator[1, S[n]]]
```

Out[∘]= 0

Look at the integer roots of the leading coefficient.

```
In[∘]:= Select[n /. Solve[LeadingCoefficient[cnn1] == 0, n], IntegerQ]
Out[∘]= {-4}
```

Check the first few (more than necessary) initial values.

```
In[∘]:= Table[matc[n, n - 1], {n, 9}]
Out[∘]= {1, 1, 1, 1, 1, 1, 1, 1, 1}
```

Proof of (H2)

Include the variable i into annc.

```
In[∘]:= ClearAll[n, j, i];

annci = ToOrePolynomial[Prepend[annc, S[i] - 1], OreAlgebra[S[n], S[j], S[i]]];
```

Annihilator for $a_{i,j} * c_{n,j}$. Recall that $a_{i,j}$ is split into two parts $a1_{i,j} + a2_{i,j}$. **No need to execute the following codes again.** Instead, import the data directly.

```
In[∘]:= start = CurrentDate[];

annH2Smnd1 = DFiniteTimesHyper[annci, mata1[i, j]];
annH2Smnd2 = DFiniteTimesHyper[annci, mata2[i, j]];
Export["annH2Smnd1.txt", {annH2Smnd1}]
Export["annH2Smnd2.txt", {annH2Smnd2}]

Print["Time used: ", CurrentDate[] - start];
Out[∘]= annH2Smnd1.txt
Out[∘]= annH2Smnd2.txt

Time used:  18.8861 s
```

$a1_{i,j} * c_{n,j}$

Import the annihilator for $a1_{i,j} * c_{n,j}$.

```
In[∘]:= annH2Smnd1 = ToExpression[Import["annH2Smnd1.txt"]];
AnnInfo[annH2Smnd1]
```

ByteCount: 3 988 184

Support:

$\{\{S_i, 1\}, \{S_n^2, S_n S_j, S_j^2, S_n, S_j, 1\}, \{S_j^3, S_n S_j, S_j^2, S_n, S_j, 1\}, \{S_n S_j^2, S_n S_j, S_j^2, S_n, S_j, 1\}\}$

degree {n, j, i}: {{0, 1, 1}, {16, 17, 8}, {9, 20, 12}, {7, 10, 8}}

Standard Monomials: $\{1, S_j, S_n, S_j^2, S_n S_j\}$

Holonomic Rank: 5

Import the 1st telescoper for $a1_{i,j} * c_{n,j}$.

```
In[∘]:= annH2CT1No1 = ToExpression[Import["annH2CT1No1.txt"]];

In[∘]:= AnnInfo[annH2CT1No1]
```

```
ByteCount: 27 393 808
Support: {{Si^5, Si^4 Sn, Si^3 Sn^2, Si^2 Sn^3, Si Sn^4, Sn^5,
   Si^4, Si^3 Sn, Si^2 Sn^2, Si Sn^3, Sn^4, Si^3, Si^2 Sn, Si Sn^2, Sn^3, Si^2, Si Sn, Sn^2, Si, Sn, 1}}
degree {i, n}: {{64, 86}}
Standard Monomials: ∞
Holonomic Rank: 1
```

```
In[]:= deltaH2CT1No1 = ToExpression[Import["deltaH2CT1No1.txt"]];

In[]:= ByteCount[deltaH2CT1No1]

Out[]= 1 777 395 424
```

Verify the 1st telescoper for $a1_{i,j} * c_{n,j}$. **Note that this step is VERY Memory-consuming, so we executed the command on an Amazon AWS cluster.** To illustrate the process, we may nonrigorously substitute the parameters with specific values.

```
In[]:= Timing[OreReduce[
         MapThread[(#1 + (S[j] - 1) ** #2) &, {annH2CT1No1, deltaH2CT1No1}], annH2Smnd1]]

Out[]= $Aborted

In[]:= subs = {n → 23, i → 135};
       {annH2CT1No1subs, deltaH2CT1No1subs} =
         OrePolynomialSubstitute[#, subs] & /@ {annH2CT1No1, deltaH2CT1No1};
       Timing[OreReduce[MapThread[(#1 + (S[j] - 1) ** #2) &,
          {annH2CT1No1subs, deltaH2CT1No1subs}], annH2Smnd1, OrePolynomialSubstitute → subs]]

Out[]= {25.4063, {0}}

In[]:= subs = {n → 511, i → 100};
       {annH2CT1No1subs, deltaH2CT1No1subs} =
         OrePolynomialSubstitute[#, subs] & /@ {annH2CT1No1, deltaH2CT1No1};
       Timing[OreReduce[MapThread[(#1 + (S[j] - 1) ** #2) &,
          {annH2CT1No1subs, deltaH2CT1No1subs}], annH2Smnd1, OrePolynomialSubstitute → subs]]

Out[]= {24.1875, {0}}
```

Import the 2nd telescoper for $a1_{i,j} * c_{n,j}$.

```
In[]:= annH2CT1No2 = ToExpression[Import["annH2CT1No2.txt"]];

In[]:= AnnInfo[annH2CT1No2]

ByteCount: 21 764 496
Support: {{Si^6, Si^5, Si^4 Sn, Si^3 Sn^2, Si^2 Sn^3, Si Sn^4,
   Si^4, Si^3 Sn, Si^2 Sn^2, Si Sn^3, Sn^4, Si^3, Si^2 Sn, Si Sn^2, Sn^3, Si^2, Si Sn, Sn^2, Si, Sn, 1}}
degree {i, n}: {{58, 74}}
Standard Monomials: ∞
Holonomic Rank: 1

In[]:= deltaH2CT1No2 = ToExpression[Import["deltaH2CT1No2.txt"]];

In[]:= ByteCount[deltaH2CT1No2]

Out[]= 1 033 241 824
```

Verify the 2nd telescoper for $a1_{i,j} * c_{n,j}$. **Note that this step is VERY Memory-consuming, so we executed the command on an Amazon AWS cluster.** To illustrate the process, we may nonrigorously substitute the parameters with specific values.

In[•]:=
```
Timing[OreReduce[
  MapThread[(#1 + (S[j] - 1) ** #2) &, {annH2CT1No2, deltaH2CT1No2}], annH2Smnd1]]
```
Out[•]= $Aborted

In[•]:=
```
subs = {n → 23, i → 135};
{annH2CT1No2subs, deltaH2CT1No2subs} =
  OrePolynomialSubstitute[#, subs] & /@ {annH2CT1No2, deltaH2CT1No2};
Timing[OreReduce[MapThread[(#1 + (S[j] - 1) ** #2) &,
   {annH2CT1No2subs, deltaH2CT1No2subs}], annH2Smnd1, OrePolynomialSubstitute → subs]]
```
Out[•]= {16.9219, {0}}

In[•]:=
```
subs = {n → 511, i → 100};
{annH2CT1No2subs, deltaH2CT1No2subs} =
  OrePolynomialSubstitute[#, subs] & /@ {annH2CT1No2, deltaH2CT1No2};
Timing[OreReduce[MapThread[(#1 + (S[j] - 1) ** #2) &,
   {annH2CT1No2subs, deltaH2CT1No2subs}], annH2Smnd1, OrePolynomialSubstitute → subs]]
```
Out[•]= {17.8594, {0}}

$a2_{i,j} * c_{n,j}$

Import the annihilator for $a2_{i,j} * c_{n,j}$.

In[•]:=
```
annH2Smnd2 = ToExpression[Import["annH2Smnd2.txt"]];
AnnInfo[annH2Smnd2]
```

ByteCount: 3 960 912

Support:

$\{\{S_i, 1\}, \{S_n^2, S_n S_j, S_j^2, S_n, S_j, 1\}, \{S_j^3, S_n S_j, S_j^2, S_n, S_j, 1\}, \{S_n S_j^2, S_n S_j, S_j^2, S_n, S_j, 1\}\}$

degree {n, j, i}: {{0, 1, 1}, {16, 17, 8}, {9, 20, 12}, {7, 10, 8}}

Standard Monomials: $\{1, S_j, S_n, S_j^2, S_n S_j\}$

Holonomic Rank: 5

Import the 1st telescoper for $a2_{i,j} * c_{n,j}$.

In[•]:=
```
annH2CT2No1 = ToExpression[Import["annH2CT2No1.txt"]];
```

In[•]:=
```
AnnInfo[annH2CT2No1]
```

ByteCount: 27 393 808

Support: $\{\{S_i^5, S_i^4 S_n, S_i^3 S_n^2, S_i^2 S_n^3, S_i S_n^4, S_n^5,$
$S_i^4, S_i^3 S_n, S_i^2 S_n^2, S_i S_n^3, S_n^4, S_i^3, S_i^2 S_n, S_i S_n^2, S_n^3, S_i^2, S_i S_n, S_n^2, S_i, S_n, 1\}\}$

degree {i, n}: {{64, 86}}

Standard Monomials: ∞

Holonomic Rank: 1

In[•]:=
```
deltaH2CT2No1 = ToExpression[Import["deltaH2CT2No1.txt"]];
```

In[•]:=
```
ByteCount[deltaH2CT2No1]
```
Out[•]= 2 657 465 696

Verify the 1st telescoper for $a2_{i,j} * c_{n,j}$. **Note that this step is VERY Memory-consuming, so we executed the command on an Amazon AWS cluster.** To illustrate the process, we may nonrigorously substitute the parameters with specific values.

In[•]:=
```
Timing[OreReduce[
  MapThread[(#1 + (S[j] - 1) ** #2) &, {annH2CT2No1, deltaH2CT2No1}], annH2Smnd2]]
```

Out[ ]= $Aborted

```
In[ ]:= subs = {n → 23, i → 135};
{annH2CT2No1subs, deltaH2CT2No1subs} =
  OrePolynomialSubstitute[#, subs] & /@ {annH2CT2No1, deltaH2CT2No1};
Timing[OreReduce[MapThread[(#1 + (S[j] - 1) ** #2) &,
    {annH2CT2No1subs, deltaH2CT2No1subs}], annH2Smnd2, OrePolynomialSubstitute → subs]]
```

Out[ ]= {23.9844, {0}}

```
In[ ]:= subs = {n → 511, i → 100};
{annH2CT2No1subs, deltaH2CT2No1subs} =
  OrePolynomialSubstitute[#, subs] & /@ {annH2CT2No1, deltaH2CT2No1};
Timing[OreReduce[MapThread[(#1 + (S[j] - 1) ** #2) &,
    {annH2CT2No1subs, deltaH2CT2No1subs}], annH2Smnd2, OrePolynomialSubstitute → subs]]
```

Out[ ]= {22.5469, {0}}

Import the 2nd telescoper for a2$_{i,j}$ * $c_{n,j}$.

```
In[ ]:= annH2CT2No2 = ToExpression[Import["annH2CT2No2.txt"]];
```

```
In[ ]:= AnnInfo[annH2CT2No2]
```

ByteCount: 20 869 424

Support: $\{\{S_i^6, S_i^5, S_i^4 S_n, S_i^3 S_n^2, S_i^2 S_n^3, S_i S_n^4, S_i^4, S_i^3 S_n, S_i^2 S_n^2, S_i S_n^3, S_n^4, S_i^3, S_i^2 S_n, S_i S_n^2, S_n^3, S_i^2, S_i S_n, S_n^2, S_i, S_n, 1\}\}$

degree {i, n}: {{56, 74}}

Standard Monomials: ∞

Holonomic Rank: 1

```
In[ ]:= deltaH2CT2No2 = ToExpression[Import["deltaH2CT2No2.txt"]];
```

```
In[ ]:= ByteCount[deltaH2CT2No2]
```

Out[ ]= 1 558 607 768

Verify the 2nd telescoper for a2$_{i,j}$ * $c_{n,j}$. **Note that this step is VERY Memory-consuming, so we executed the command on an Amazon AWS cluster.** To illustrate the process, we may nonrigorously substitute the parameters with specific values.

```
In[ ]:= Timing[OreReduce[
  MapThread[(#1 + (S[j] - 1) ** #2) &, {annH2CT2No2, deltaH2CT2No2}], annH2Smnd2]]
```

Out[ ]= $Aborted

```
In[ ]:= subs = {n → 23, i → 135};
{annH2CT2No2subs, deltaH2CT2No2subs} =
  OrePolynomialSubstitute[#, subs] & /@ {annH2CT2No2, deltaH2CT2No2};
Timing[OreReduce[MapThread[(#1 + (S[j] - 1) ** #2) &,
    {annH2CT2No2subs, deltaH2CT2No2subs}], annH2Smnd2, OrePolynomialSubstitute → subs]]
```

Out[ ]= {18.1406, {0}}

```
In[ ]:= subs = {n → 511, i → 100};
{annH2CT2No2subs, deltaH2CT2No2subs} =
  OrePolynomialSubstitute[#, subs] & /@ {annH2CT2No2, deltaH2CT2No2};
Timing[OreReduce[MapThread[(#1 + (S[j] - 1) ** #2) &,
    {annH2CT2No2subs, deltaH2CT2No2subs}], annH2Smnd2, OrePolynomialSubstitute → subs]]
```

Out[ ]= {17.2656, {0}}

$a_{i,j} * c_{n,j}$

Check that annH2CT1No1 and annH2CT2No1 differ by a scalar. So annH2CT1No1 also annihilates $\sum_j a2_{i,j} * c_{n,j}$.

```
In[ ]:= annH2CT1No1LeadingCoefficient = LeadingCoefficient[annH2CT1No1[[1]]];
annH2CT2No1LeadingCoefficient = LeadingCoefficient[annH2CT2No1[[1]]];
ScalarNo1 = Factor[annH2CT1No1LeadingCoefficient / annH2CT2No1LeadingCoefficient]
Out[ ]= 1

In[ ]:= annH2CT1No1CoeffList = OrePolynomialListCoefficients[annH2CT1No1[[1]]];
annH2CT2No1CoeffList = OrePolynomialListCoefficients[annH2CT2No1[[1]]];

In[ ]:= Expand[annH2CT1No1CoeffList] == Expand[ScalarNo1 * annH2CT2No1CoeffList]
Out[ ]= True
```

Check that annH2CT1No2 and annH2CT2No2 differ by a scalar. So annH2CT1No2 also annihilates $\sum_j a2_{i,j} * c_{n,j}$.

```
In[ ]:= annH2CT1No2LeadingCoefficient = LeadingCoefficient[annH2CT1No2[[1]]];
annH2CT2No2LeadingCoefficient = LeadingCoefficient[annH2CT2No2[[1]]];
ScalarNo2 = Factor[annH2CT1No2LeadingCoefficient / annH2CT2No2LeadingCoefficient]
Out[ ]= (1 + i) (6 + i)

In[ ]:= annH2CT1No2CoeffList = OrePolynomialListCoefficients[annH2CT1No2[[1]]];
annH2CT2No2CoeffList = OrePolynomialListCoefficients[annH2CT2No2[[1]]];

In[ ]:= Expand[annH2CT1No2CoeffList] == Expand[ScalarNo2 * annH2CT2No2CoeffList]
Out[ ]= True

In[ ]:= annH2CTNo1 = annH2CT1No1;
annH2CTNo2 = annH2CT1No2;
```

Determine the singularities with $i \geq 5$.

```
In[ ]:= LeadingExponent[annH2CTNo1[[1]]]
Out[ ]= {5, 0}

In[ ]:= LeadingExponent[annH2CTNo2[[1]]]
Out[ ]= {6, 0}

In[ ]:= coeff1 = LeadingCoefficient[annH2CTNo1[[1]]] /.
    {i → i - LeadingExponent[annH2CTNo1[[1]]][[1]],
     n → n - LeadingExponent[annH2CTNo1[[1]]][[2]]};

coeff2 = LeadingCoefficient[annH2CTNo2[[1]]] /.
    {i → i - LeadingExponent[annH2CTNo2[[1]]][[1]],
     n → n - LeadingExponent[annH2CTNo2[[1]]][[2]]};
```

Check the singularities at $i = 5$.

```
In[ ]:= Solve[(coeff1 /. {i → 5}) == 0 && 5 < n - 1, n, Integers]
Out[ ]= {}
```

Check the singularities at $i \geq 6$.

The following code is too Time- and especially Memory-consuming so we finally executed it on an Amazon cluster. The output is empty { }. (To convince the reader, we also perform some numerical verification.)

```
In[•]:= Solve[coeff1 == 0 && coeff2 == 0 && i ≥ 6 && i < n - 1, {n, i}, Integers]
Out[•]= $Aborted
```

```
In[•]:= sol = {};
        For[ii = 6, ii ≤ 100, ii++,
          For[nn = ii + 2, nn ≤ ii + 100, nn++,
            If[(coeff1 /. {n → nn, i → ii}) == 0 && (coeff2 /. {n → nn, i → ii}) == 0,
              AppendTo[sol, {n → nn, i → ii}];
             ];
           ];
         ];
        sol
Out[•]= {}
```

Check values at *i* = 0.

```
In[•]:= ii = 0;
```

Apply creative telescoping to $a1_{i,j} * c_{n,j}$.

```
In[•]:= annii = DFiniteTimesHyper[annc, mata1[ii, j]];
        {anniiCT, deltaiiCT} = FindCreativeTelescoping[annii, S[j] - 1];
```

The telescoper part is 1. We then find that the inhomogeneous part is **nonvanishing**, and hence the analysis becomes trickier.

```
In[•]:= anniiCT
Out[•]= {1}
```

Here we note that the delta part is supported on a collection of power products $S[n]^a\, S[j]^b$ with, in particular, $0 \leq b \leq 2$.

```
In[•]:= Support[deltaiiCT[[1]]]
Out[•]= {{Sn Sj, Sj^2, Sn, Sj, 1}}
```

```
In[•]:= deltaiiCT[[1, 1]][[1, All, 2]]
        BB = Max[%[[All, 2]]]
Out[•]= {{1, 1}, {0, 2}, {1, 0}, {0, 1}, {0, 0}}
Out[•]= 2
```

We write explicitly the inhomogeneous part by calling $c_{n,j}$ temporarily under the name ff[n,j].

```
In[•]:= ClearAll[ff];
        inhomii = ApplyOreOperator[deltaiiCT[[1]], mata1[ii, j] * ff[n, j]][[1]];
```

The inhomogeneous part at $j = n$ vanishes by recalling that $c_{n,n-1} = 1$ and $c_{n,j} = 0$ when $j \geq n$.

```
In[•]:= sumiin = (inhomii /. {j → n}) /.
          {(ff[a_, b_] /; (SameQ[a - b, 1])) → 1, (ff[a_, b_] /; (a - b ≤ 0)) → 0}
```

Out[∘]= 0

The inhomogeneous part at $j = 0$ will be split into parts by collecting $c_{?,b}$.

```
In[∘]:= sumii = inhomii /. {j → 0};

In[∘]:= ClearAll[part];
For[bb = 0, bb ≤ BB, bb++,
   part[bb] = sumii /. {(ff[a_, b_] /; SameQ[b, bb] == False) → 0};
  ];
```

We shall see that by combining these parts, our $\Sigma_{i,n}^{(1)}$ will be recovered.

```
In[∘]:= MAX = 3;
For[bb = 0, bb ≤ BB, bb++,
   Print["Part ", bb, ":\n",
     Table[part[bb] /. {n → nn, ff → matc}, {nn, ii + 2, ii + 2 + MAX}], "\n=========="];
  ];
Print["Total:\n", Table[Sum[part[bb] /. {n → nn, ff → matc}, {bb, 0, BB}],
    {nn, ii + 2, ii + 2 + MAX}], "\n====================\n",
   "Compare with ∑_j", Subscript["a1", ToString[ii] <> ",j"], "c_{n,j}:\n",
   Table[Sum[mata1[ii, j] * matc[nn, j], {j, 0, nn - 1}], {nn, ii + 2, ii + 2 + MAX}]];
```

Part 0:

$\left\{\frac{17}{52}, \frac{6206}{4845}, -\frac{173279}{100625}, \frac{461669704}{230166475}\right\}$

==========

Part 1:

$\left\{-\frac{17}{52}, -\frac{164677}{155040}, \frac{10620081}{6440000}, -\frac{14637479253}{7365327200}\right\}$

==========

Part 2:

$\left\{0, -\frac{7}{32}, \frac{817}{11200}, -\frac{6049}{327712}\right\}$

==========

Total:

{0, 0, 0, 0}

====================

Compare with $\sum_j a1_{0,j} c_{n,j}$:

{0, 0, 0, 0}

```
In[∘]:= MAX = 10;
Table[Sum[mata1[ii, j] * matc[nn, j], {j, 0, nn - 1}], {nn, ii + 2, ii + MAX}] ==
  Table[Sum[part[bb] /. {n → nn, ff → matc}, {bb, 0, BB}], {nn, ii + 2, ii + MAX}]

Out[∘]= True
```

Derive separately the annihilating ideal anncii[b] for the univariate sequence $c_{n,j}$ with fixed $j = b$.

```
In[∘]:= ClearAll[anncii];
For[bb = 0, bb ≤ BB, bb++,
   anncii[bb] = DFiniteSubstitute[annc, {j → bb}];
  ];
```

In what follows, we show that $\Sigma_{i,n}^{(1)}$ is annihilated by the annihilating ideal anncii[0] for $c_{n,0}$.

For each part obtained earlier, we apply the Ore polynomial anncii[0].

Putting them together, the Ore polynomial anncii[0] is indeed applied to $\Sigma_{i,n}^{(1)}$; the annihilation is illustrated numerically below.

In[ ]:=
```
ClearAll[partunderannc0];
MAX = 15;
For[bb = 0, bb ≤ BB, bb++,
  partunderannc0[bb] = ApplyOreOperator[anncii[0][[1]], part[bb]];
 ];
Table[Sum[partunderannc0[bb] /. {n → nn, ff → matc}, {bb, 0, BB}], {nn, ii + 2, ii + MAX}]
```

Out[ ]=
```
{0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0}
```

Note that the b-th part acted by anncii[0] can be rewritten as the univariate sequence $c_{n,b}$ acted by a certain Ore polynomial O_b in the algebra $S[n]$. Now we derive an annihilating ideal for each $O_b \cdot c_{n,b}$.

In[ ]:=
```
ClearAll[orepolypart, annpart];
For[bb = 0, bb ≤ BB, bb++,
  orepolypart[bb] = ToOrePolynomial[partunderannc0[bb], ff[n, bb]];
  annpart[bb] = DFiniteOreAction[anncii[bb], orepolypart[bb]];
 ];
```

We can indeed check the correctness of these annihilating ideals numerically.

In[ ]:=
```
ClearAll[gg];
ClearAll[recpart, valpart];
MAX = 15;
For[bb = 0, bb ≤ BB, bb++,
  recpart[bb] = ApplyOreOperator[annpart[bb], gg[n]][[1]];
  valpart[bb][nn_] := partunderannc0[bb] /. {n → nn, ff → matc};
  Print["Part ", bb, ":\n", Table[
    recpart[bb] /. {n → nn, gg → valpart[bb]}, {nn, ii + 2, ii + MAX}], "\n=========="];
 ];
```

```
Part 0:
{0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0}
==========
Part 1:
{0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0}
==========
Part 2:
{0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0}
==========
```

By the closure property, we obtain an annihilating ideal for $\sum_b O_b \cdot c_{n,b} = \text{anncii}[0] \cdot \Sigma_{i,n}^{(1)}$. We then compute the singularities.

In[ ]:=
```
anntotal = DFinitePlus[Sequence @@ Table[annpart[bb], {bb, 0, BB}]];
```

In[ ]:=
```
LeadingExponent[anntotal[[1]]][[1]]
```

Out[ ]=
```
12
```

In[ ]:=
```
Solve[
 (LeadingCoefficient[anntotal[[1]]] /. {n → n - LeadingExponent[anntotal[[1]]][[1]]}) ==
  0, n, Integers]
```

Out[ ]=
```
{{n → -7}, {n → -7}, {n → -6}, {n → -6}, {n → -3}, {n → -2}, {n → -1}, {n → 0}, {n → 1}}
```

We check that the initial values for $\sum_b O_b \cdot c_{n,b} = \text{anncii}[0] \cdot \Sigma_{i,n}^{(1)}$ are all 0, and hence they vanish for all

$n \geq i+2$. **This confirms that $\Sigma_{i,n}^{(1)}$ is annihilated by anncii[0].**

```
In[•]:= Table[Sum[partunderannc0[bb] /. {n → nn, ff → matc}, {bb, 0, BB}],
         {nn, ii + 2, ii + 2 + LeadingExponent[anntotal[[1]]][[1]]}]

        ClearAll[sigma1, hh];
        sigma[nn_] := Sum[mata1[ii, j] * matc[nn, j], {j, 0, nn - 1}];
        sigma1underannc0 = ApplyOreOperator[anncii[0][[1]], hh[n]];
        Table[sigma1underannc0 /. {n → nn, hh → sigma},
         {nn, ii + 2, ii + 2 + LeadingExponent[anntotal[[1]]][[1]]}]
Out[•]= {0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0}
Out[•]= {0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0}
```

Note that $a2_{i,j}$ equals Binomial[$-i$, 2] when $j=0$, and 0 when $j \geq 1$ for our choice of i. **Hence, $\Sigma_{i,n}^{(2)} = \sum_{j=0}^{n-1} a2_{i,j}\, c_{n,j}$ = Binomial[$-i$, 2]$* c_{n,0}$ is also annihilated by anncii[0].**

Consequently, $\Sigma_{i,n}$ is annihilated by anncii[0].

Finally, after computing the singularities, we check that $\Sigma_{i,n}$ vanishes for its initial values and hence for all $n \geq i+2$.

```
In[•]:= LeadingExponent[anncii[0][[1]]][[1]]
Out[•]= 5

In[•]:= Solve[(LeadingCoefficient[anncii[0][[1]]] /.
            {n → n - LeadingExponent[anncii[0][[1]]][[1]]}) == 0, n, Integers]
Out[•]= {{n → 0}, {n → 1}}

In[•]:= Table[Sum[mata[ii, j] * matc[n, j], {j, 0, n - 1}],
         {n, ii + 2, ii + 2 + LeadingExponent[anncii[0][[1]]][[1]]}]
Out[•]= {0, 0, 0, 0, 0, 0}
```

Check values at $i = 1$.

```
In[•]:= ii = 1;
```

Apply creative telescoping to $a1_{i,j} * c_{n,j}$.

```
In[•]:= annii = DFiniteTimesHyper[annc, mata1[ii, j]];
        {anniiCT, deltaiiCT} = FindCreativeTelescoping[annii, S[j] - 1];
```

The telescoper part is 1. We then find that the inhomogeneous part is **nonvanishing**, and hence the analysis becomes trickier.

```
In[•]:= anniiCT
Out[•]= {1}
```

Here we note that the delta part is supported on a collection of power products $S[n]^a\, S[j]^b$ with, in particular, $0 \leq b \leq 2$.

```
In[•]:= Support[deltaiiCT[[1]]]
Out[•]= {{Sn Sj, Sj^2, Sn, Sj, 1}}

In[•]:= deltaiiCT[[1, 1]][[1, All, 2]]
        BB = Max[%[[All, 2]]]
```

Out[∘]= {{1, 1}, {0, 2}, {1, 0}, {0, 1}, {0, 0}}

Out[∘]= 2

We write explicitly the inhomogeneous part by calling $c_{n,j}$ temporarily under the name ff[n,j].

```
In[∘]:= ClearAll[ff];
inhomii = ApplyOreOperator[deltaiiCT[[1]], mata1[ii, j] * ff[n, j]][[1]];
```

The inhomogeneous part at $j = n$ vanishes by recalling that $c_{n,n-1} = 1$ and $c_{n,j} = 0$ when $j \geq n$.

```
In[∘]:= sumiin = (inhomii /. {j → n}) /.
   {(ff[a_, b_] /; (SameQ[a - b, 1])) → 1, (ff[a_, b_] /; (a - b ≤ 0)) → 0}
Out[∘]= 0
```

The inhomogeneous part at $j = 0$ will be split into parts by collecting $c_{?,b}$.

```
In[∘]:= sumii = inhomii /. {j → 0};

In[∘]:= ClearAll[part];
For[bb = 0, bb ≤ BB, bb++,
   part[bb] = sumii /. {(ff[a_, b_] /; SameQ[b, bb] == False) → 0};
  ];
```

We shall see that by combining these parts, our $\Sigma^{(1)}_{i,n}$ will be recovered.

```
In[∘]:= MAX = 3;
For[bb = 0, bb ≤ BB, bb++,
   Print["Part ", bb, ":\n",
     Table[part[bb] /. {n → nn, ff → matc}, {nn, ii + 2, ii + 2 + MAX}], "\n=========="];
  ];
Print["Total:\n", Table[Sum[part[bb] /. {n → nn, ff → matc}, {bb, 0, BB}],
    {nn, ii + 2, ii + 2 + MAX}], "\n====================\n",
   "Compare with ∑_j", Subscript["a1", ToString[ii] <> ",j"], "c_{n,j}:\n",
   Table[Sum[mata1[ii, j] * matc[nn, j], {j, 0, nn - 1}], {nn, ii + 2, ii + 2 + MAX}]];
```

Part 0:

$\left\{-\frac{361387}{14535}, \frac{491577}{201250}, \frac{1005819904}{230166475}, -\frac{3596392}{454545}\right\}$

==========

Part 1:

$\left\{\frac{14358011}{465120}, -\frac{17988489}{6440000}, -\frac{116891101609}{22095981600}, \frac{166965473}{18181800}\right\}$

==========

Part 2:

$\left\{-\frac{679}{96}, \frac{16727}{11200}, -\frac{299557}{983136}, \frac{2933}{70200}\right\}$

==========

Total:

$\left\{-\frac{16}{15}, \frac{8}{7}, -\frac{256}{209}, \frac{256}{195}\right\}$

====================

Compare with $\sum_j a1_{1,j} c_{n,j}$:

$\left\{-\frac{16}{15}, \frac{8}{7}, -\frac{256}{209}, \frac{256}{195}\right\}$

In[∘]:=
```
MAX = 10;
Table[Sum[mata1[ii, j] * matc[nn, j], {j, 0, nn - 1}], {nn, ii + 2, ii + MAX}] ==
 Table[Sum[part[bb] /. {n → nn, ff → matc}, {bb, 0, BB}], {nn, ii + 2, ii + MAX}]
```
Out[∘]=
```
True
```

Derive separately the annihilating ideal anncii[b] for the univariate sequence $c_{n,j}$ with fixed $j = b$.

In[∘]:=
```
ClearAll[anncii];
For[bb = 0, bb ≤ BB, bb++,
  anncii[bb] = DFiniteSubstitute[annc, {j → bb}];
 ];
```

In what follows, we show that $\Sigma_{i,n}^{(1)}$ is annihilated by the annihilating ideal anncii[0] for $c_{n,0}$.

For each part obtained earlier, we apply the Ore polynomial anncii[0].

Putting them together, the Ore polynomial anncii[0] is indeed applied to $\Sigma_{i,n}^{(1)}$; the annihilation is illustrated numerically below.

In[∘]:=
```
ClearAll[partunderannc0];
MAX = 15;
For[bb = 0, bb ≤ BB, bb++,
  partunderannc0[bb] = ApplyOreOperator[anncii[0][[1]], part[bb]];
 ];
Table[Sum[partunderannc0[bb] /. {n → nn, ff → matc}, {bb, 0, BB}], {nn, ii + 2, ii + MAX}]
```
Out[∘]=
```
{0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0}
```

Note that the b-th part acted by anncii[0] can be rewritten as the univariate sequence $c_{n,b}$ acted by a certain Ore polynomial O_b in the algebra $S[n]$. Now we derive an annihilating ideal for each $O_b \cdot c_{n,b}$.

In[∘]:=
```
ClearAll[orepolypart, annpart];
For[bb = 0, bb ≤ BB, bb++,
  orepolypart[bb] = ToOrePolynomial[partunderannc0[bb], ff[n, bb]];
  annpart[bb] = DFiniteOreAction[anncii[bb], orepolypart[bb]];
 ];
```

We can indeed check the correctness of these annihilating ideals numerically.

In[∘]:=
```
ClearAll[gg];
ClearAll[recpart, valpart];
MAX = 15;
For[bb = 0, bb ≤ BB, bb++,
  recpart[bb] = ApplyOreOperator[annpart[bb], gg[n]][[1]];
  valpart[bb][nn_] := partunderannc0[bb] /. {n → nn, ff → matc};
  Print["Part ", bb, ":\n", Table[
    recpart[bb] /. {n → nn, gg → valpart[bb]}, {nn, ii + 2, ii + MAX}], "\n=========="];
 ];
```

```
Part 0:
{0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0}
==========

Part 1:
{0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0}
==========
```

```
Part 2:
{0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0}
==========
```

By the closure property, we obtain an annihilating ideal for $\sum_b O_b \cdot c_{n,b}$ = anncii[0]$\cdot \Sigma_{i,n}^{(1)}$. We then compute the singularities.

```
In[∘]:= anntotal = DFinitePlus[Sequence @@ Table[annpart[bb], {bb, 0, BB}]];

In[∘]:= LeadingExponent[anntotal[[1]]][[1]]

Out[∘]= 12

In[∘]:= Solve[
         (LeadingCoefficient[anntotal[[1]]] /. {n → n - LeadingExponent[anntotal[[1]]][[1]]}) ==
          0, n, Integers]

Out[∘]= {{n → -7}, {n → -7}, {n → -6}, {n → -6}, {n → -3}, {n → -2}}
```

We check that the initial values for $\sum_b O_b \cdot c_{n,b}$ = anncii[0]$\cdot \Sigma_{i,n}^{(1)}$ are all 0, and hence they vanish for all $n \geq i+2$. **This confirms that $\Sigma_{i,n}^{(1)}$ is annihilated by anncii[0].**

```
In[∘]:= Table[Sum[partunderannc0[bb] /. {n → nn, ff → matc}, {bb, 0, BB}],
         {nn, ii + 2, ii + 2 + LeadingExponent[anntotal[[1]]][[1]]}]

        ClearAll[sigma1, hh];
        sigma[nn_] := Sum[mata1[ii, j] * matc[nn, j], {j, 0, nn - 1}];
        sigma1underannc0 = ApplyOreOperator[anncii[0][[1]], hh[n]];
        Table[sigma1underannc0 /. {n → nn, hh → sigma},
         {nn, ii + 2, ii + 2 + LeadingExponent[anntotal[[1]]][[1]]}]

Out[∘]= {0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0}

Out[∘]= {0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0}
```

Note that $a2_{i,j}$ equals Binomial[$-i$, 2] when $j = 0$, and 0 when $j \geq 1$ for our choice of i. **Hence, $\Sigma_{i,n}^{(2)} = \sum_{j=0}^{n-1} a2_{i,j}\, c_{n,j}$ = Binomial[$-i$, 2]$* c_{n,0}$ is also annihilated by anncii[0].**

Consequently, $\Sigma_{i,n}$ is annihilated by anncii[0].

Finally, after computing the singularities, we check that $\Sigma_{i,n}$ vanishes for its initial values and hence for all $n \geq i+2$.

```
In[∘]:= LeadingExponent[anncii[0][[1]]][[1]]

Out[∘]= 5

In[∘]:= Solve[(LeadingCoefficient[anncii[0][[1]]] /.
            {n → n - LeadingExponent[anncii[0][[1]]][[1]]}) == 0, n, Integers]

Out[∘]= {{n → 0}, {n → 1}}

In[∘]:= Table[Sum[mata[ii, j] * matc[n, j], {j, 0, n - 1}],
         {n, ii + 2, ii + 2 + LeadingExponent[anncii[0][[1]]][[1]]}]

Out[∘]= {0, 0, 0, 0, 0, 0}
```

Check values at $i = 2$.

```
In[∘]:= ii = 2;
```

Apply creative telescoping to $a1_{i,j} * c_{n,j}$.

```
In[•]:= annii = DFiniteTimesHyper[annc, mata1[ii, j]];
{anniiCT, deltaiiCT} = FindCreativeTelescoping[annii, S[j] - 1];
```

The telescoper part is 1. We then find that the inhomogeneous part is **nonvanishing**, and hence the analysis becomes trickier.

```
In[•]:= anniiCT
Out[•]= {1}
```

Here we note that the delta part is supported on a collection of power products $S[n]^a\ S[j]^b$ with, in particular, $0 \leq b \leq 2$.

```
In[•]:= Support[deltaiiCT[[1]]]
Out[•]= {{Sn Sj, Sj^2, Sn, Sj, 1}}
```

```
In[•]:= deltaiiCT[[1, 1]][[1, All, 2]]
BB = Max[%[[All, 2]]]
Out[•]= {{1, 1}, {0, 2}, {1, 0}, {0, 1}, {0, 0}}
Out[•]= 2
```

We write explicitly the inhomogeneous part by calling $c_{n,j}$ temporarily under the name ff[n,j].

```
In[•]:= ClearAll[ff];
inhomii = ApplyOreOperator[deltaiiCT[[1]], mata1[ii, j] * ff[n, j]][[1]];
```

The inhomogeneous part at $j = n$ vanishes by recalling that $c_{n,n-1} = 1$ and $c_{n,j} = 0$ when $j \geq n$.

```
In[•]:= sumiin = (inhomii /. {j → n}) /.
   {(ff[a_, b_] /; (SameQ[a - b, 1])) → 1, (ff[a_, b_] /; (a - b ≤ 0)) → 0}
Out[•]= 0
```

The inhomogeneous part at $j = 0$ will be split into parts by collecting $c_{?,b}$.

```
In[•]:= sumii = inhomii /. {j → 0};
```

```
In[•]:= ClearAll[part];
For[bb = 0, bb ≤ BB, bb++,
   part[bb] = sumii /. {(ff[a_, b_] /; SameQ[b, bb] == False) → 0};
  ];
```

We shall see that by combining these parts, our $\Sigma^{(1)}_{i,n}$ will be recovered.

```
In[•]:= MAX = 3;
For[bb = 0, bb ≤ BB, bb++,
   Print["Part ", bb, ":\n",
     Table[part[bb] /. {n → nn, ff → matc}, {nn, ii + 2, ii + 2 + MAX}], "\n=========="];
  ];
Print["Total:\n", Table[Sum[part[bb] /. {n → nn, ff → matc}, {bb, 0, BB}],
    {nn, ii + 2, ii + 2 + MAX}], "\n====================\n",
   "Compare with ∑j", Subscript["a1", ToString[ii] <> ",j"], "cn,j:\n",
   Table[Sum[mata1[ii, j] * matc[nn, j], {j, 0, nn - 1}], {nn, ii + 2, ii + 2 + MAX}]];
```

Part 0:
$\left\{\frac{11537151}{40250}, -\frac{356208192}{4184845}, \frac{12324148}{454545}, \frac{118130388}{721146335}\right\}$
==========

Part 1:
$\left\{-\frac{389176407}{1288000}, \frac{123843952579}{1473065440}, -\frac{97570481}{4155840}, -\frac{3162901908}{721146335}\right\}$
==========

Part 2:
$\left\{\frac{6063}{320}, -\frac{861325}{327712}, \frac{238411}{786240}, 0\right\}$
==========

Total:
$\left\{\frac{24}{7}, -\frac{768}{209}, \frac{256}{65}, -\frac{49344}{11687}\right\}$
====================

Compare with $\sum_j a1_{2,j} c_{n,j}$:
$\left\{\frac{24}{7}, -\frac{768}{209}, \frac{256}{65}, -\frac{49344}{11687}\right\}$

In[∘]:=
```
MAX = 10;
Table[Sum[mata1[ii, j] * matc[nn, j], {j, 0, nn - 1}], {nn, ii + 2, ii + MAX}] ==
 Table[Sum[part[bb] /. {n → nn, ff → matc}, {bb, 0, BB}], {nn, ii + 2, ii + MAX}]
```

Out[∘]= True

Derive separately the annihilating ideal anncii[b] for the univariate sequence $c_{n,j}$ with fixed $j = b$.

In[∘]:=
```
ClearAll[anncii];
For[bb = 0, bb ≤ BB, bb++,
  anncii[bb] = DFiniteSubstitute[annc, {j → bb}];
 ];
```

In what follows, we show that $\Sigma_{i,n}^{(1)}$ is annihilated by the annihilating ideal anncii[0] for $c_{n,0}$.

For each part obtained earlier, we apply the Ore polynomial anncii[0].

Putting them together, the Ore polynomial anncii[0] is indeed applied to $\Sigma_{i,n}^{(1)}$; the annihilation is illustrated numerically below.

In[∘]:=
```
ClearAll[partunderannc0];
MAX = 15;
For[bb = 0, bb ≤ BB, bb++,
  partunderannc0[bb] = ApplyOreOperator[anncii[0][[1]], part[bb]];
 ];
Table[Sum[partunderannc0[bb] /. {n → nn, ff → matc}, {bb, 0, BB}], {nn, ii + 2, ii + MAX}]
```

Out[∘]= {0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0}

Note that the b-th part acted by anncii[0] can be rewritten as the univariate sequence $c_{n,b}$ acted by a certain Ore polynomial O_b in the algebra $S[n]$. Now we derive an annihilating ideal for each $O_b \cdot c_{n,b}$.

In[∘]:=
```
ClearAll[orepolypart, annpart];
For[bb = 0, bb ≤ BB, bb++,
  orepolypart[bb] = ToOrePolynomial[partunderannc0[bb], ff[n, bb]];
  annpart[bb] = DFiniteOreAction[anncii[bb], orepolypart[bb]];
 ];
```

We can indeed check the correctness of these annihilating ideals numerically.

In[∘]:=
```
ClearAll[gg];
ClearAll[recpart, valpart];
MAX = 15;
For[bb = 0, bb ≤ BB, bb++,
  recpart[bb] = ApplyOreOperator[annpart[bb], gg[n]][[1]];
  valpart[bb][nn_] := partunderannc0[bb] /. {n → nn, ff → matc};
  Print["Part ", bb, ":\n", Table[
    recpart[bb] /. {n → nn, gg → valpart[bb]}, {nn, ii + 2, ii + MAX}], "\n=========="];
 ];
```

```
Part 0:
{0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0}
==========
Part 1:
{0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0}
==========
Part 2:
{0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0}
==========
```

By the closure property, we obtain an annihilating ideal for $\sum_b O_b \cdot c_{n,b} = \text{anncii}[0] \cdot \Sigma_{i,n}^{(1)}$. We then compute the singularities.

In[∘]:=
```
anntotal = DFinitePlus[Sequence @@ Table[annpart[bb], {bb, 0, BB}]];
```

In[∘]:=
```
LeadingExponent[anntotal[[1]]][[1]]
```

Out[∘]= 12

In[∘]:=
```
Solve[
 (LeadingCoefficient[anntotal[[1]]] /. {n → n - LeadingExponent[anntotal[[1]]][[1]]}) ==
  0, n, Integers]
```

Out[∘]= {{n → -7}, {n → -7}, {n → -6}, {n → -6}, {n → -3}, {n → -2}, {n → -1}}

We check that the initial values for $\sum_b O_b \cdot c_{n,b} = \text{anncii}[0] \cdot \Sigma_{i,n}^{(1)}$ are all 0, and hence they vanish for all $n \geq i + 2$. **This confirms that $\Sigma_{i,n}^{(1)}$ is annihilated by anncii[0].**

In[∘]:=
```
Table[Sum[partunderannc0[bb] /. {n → nn, ff → matc}, {bb, 0, BB}],
 {nn, ii + 2, ii + 2 + LeadingExponent[anntotal[[1]]][[1]]}]

ClearAll[sigma1, hh];
sigma[nn_] := Sum[mata1[ii, j] * matc[nn, j], {j, 0, nn - 1}];
sigma1underannc0 = ApplyOreOperator[anncii[0][[1]], hh[n]];
Table[sigma1underannc0 /. {n → nn, hh → sigma},
 {nn, ii + 2, ii + 2 + LeadingExponent[anntotal[[1]]][[1]]}]
```

Out[∘]= {0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0}

Out[∘]= {0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0}

Note that $a2_{i,j}$ equals Binomial[−i, 2] when $j = 0$, and 0 when $j \geq 1$ for our choice of i. **Hence, $\Sigma^{(2)}_{i,n} = \sum_{j=0}^{n-1} a2_{i,j}\, c_{n,j}$ = Binomial[−i, 2] ∗ $c_{n,0}$ is also annihilated by anncii[0].**

Consequently, $\Sigma_{i,n}$ is annihilated by anncii[0].

Finally, after computing the singularities, we check that $\Sigma_{i,n}$ vanishes for its initial values and hence for all $n \geq i + 2$.

```
In[∘]:= LeadingExponent[anncii[0][[1]]][[1]]
Out[∘]= 5

In[∘]:= Solve[(LeadingCoefficient[anncii[0][[1]]] /.
            {n → n - LeadingExponent[anncii[0][[1]]][[1]]}) == 0, n, Integers]
Out[∘]= {{n → 0}, {n → 1}}

In[∘]:= Table[Sum[mata[ii, j] * matc[n, j], {j, 0, n - 1}],
          {n, ii + 2, ii + 2 + LeadingExponent[anncii[0][[1]]][[1]]}]
Out[∘]= {0, 0, 0, 0, 0, 0}
```

Check values at $i = 3$.

```
In[∘]:= ii = 3;
```

Apply creative telescoping to $a1_{i,j} * c_{n,j}$.

```
In[∘]:= annii = DFiniteTimesHyper[annc, mata1[ii, j]];
        {anniiCT, deltaiiCT} = FindCreativeTelescoping[annii, S[j] - 1];
```

The telescoper part is 1. We then find that the inhomogeneous part is **nonvanishing**, and hence the analysis becomes trickier.

```
In[∘]:= anniiCT
Out[∘]= {1}
```

Here we note that the delta part is supported on a collection of power products $S[n]^a\, S[j]^b$ with, in particular, $0 \leq b \leq 2$.

```
In[∘]:= Support[deltaiiCT[[1]]]
Out[∘]= {{Sn Sj, Sj^2, Sn, Sj, 1}}

In[∘]:= deltaiiCT[[1, 1]][[1, All, 2]]
        BB = Max[%[[All, 2]]]
Out[∘]= {{1, 1}, {0, 2}, {1, 0}, {0, 1}, {0, 0}}
Out[∘]= 2
```

We write explicitly the inhomogeneous part by calling $c_{n,j}$ temporarily under the name ff[n,j].

```
In[∘]:= ClearAll[ff];
        inhomii = ApplyOreOperator[deltaiiCT[[1]], mata1[ii, j] * ff[n, j]][[1]];
```

The inhomogeneous part at $j = n$ vanishes by recalling that $c_{n,n-1} = 1$ and $c_{n,j} = 0$ when $j \geq n$.

```
In[∘]:= sumiin = (inhomii /. {j → n}) /.
          {(ff[a_, b_] /; (SameQ[a - b, 1])) → 1, (ff[a_, b_] /; (a - b ≤ 0)) → 0}
Out[∘]= 0
```

The inhomogeneous part at $j = 0$ will be split into parts by collecting $c_{?,b}$.

In[∘]:=
```
sumii = inhomii /. {j → 0};
```

In[∘]:=
```
ClearAll[part];
For[bb = 0, bb ≤ BB, bb++,
  part[bb] = sumii /. {(ff[a_, b_] /; SameQ[b, bb] == False) → 0};
 ];
```

We shall see that by combining these parts, our $\Sigma_{i,n}^{(1)}$ will be recovered.

In[∘]:=
```
MAX = 3;
For[bb = 0, bb ≤ BB, bb++,
  Print["Part ", bb, ":\n",
   Table[part[bb] /. {n → nn, ff → matc}, {nn, ii + 2, ii + 2 + MAX}], "\n=========="];
 ];
Print["Total:\n", Table[Sum[part[bb] /. {n → nn, ff → matc}, {bb, 0, BB}],
   {nn, ii + 2, ii + 2 + MAX}], "\n====================\n",
  "Compare with ∑j", Subscript["a1", ToString[ii] <> ",j"], "cn,j:\n",
  Table[Sum[mata1[ii, j] * matc[nn, j], {j, 0, nn - 1}], {nn, ii + 2, ii + 2 + MAX}]];
```

Part 0:

$\left\{-\frac{363794196448}{230166475}, \frac{19663312}{34965}, -\frac{961710177546}{3605731675}, \frac{616211719052}{4715150685}\right\}$

==========

Part 1:

$\left\{\frac{5875549896493}{3682663600}, -\frac{202300741}{363636}, \frac{931262462346}{3605731675}, -\frac{114701979676}{943030137}\right\}$

==========

Part 2:

$\left\{-\frac{3644391}{163856}, \frac{18017}{9828}, 0, 0\right\}$

==========

Total:

$\left\{-\frac{1536}{209}, \frac{512}{65}, -\frac{98688}{11687}, \frac{882944}{97495}\right\}$

====================

Compare with $\sum_j a1_{3,j} c_{n,j}$:

$\left\{-\frac{1536}{209}, \frac{512}{65}, -\frac{98688}{11687}, \frac{882944}{97495}\right\}$

In[∘]:=
```
MAX = 10;
Table[Sum[mata1[ii, j] * matc[nn, j], {j, 0, nn - 1}], {nn, ii + 2, ii + MAX}] ==
 Table[Sum[part[bb] /. {n → nn, ff → matc}, {bb, 0, BB}], {nn, ii + 2, ii + MAX}]
```

Out[∘]= True

Derive separately the annihilating ideal anncii[b] for the univariate sequence $c_{n,j}$ with fixed $j = b$.

In[∘]:=
```
ClearAll[anncii];
For[bb = 0, bb ≤ BB, bb++,
  anncii[bb] = DFiniteSubstitute[annc, {j → bb}];
 ];
```

In what follows, we show that $\Sigma_{i,n}^{(1)}$ is annihilated by the annihilating ideal anncii[0] for $c_{n,0}$.

For each part obtained earlier, we apply the Ore polynomial anncii[0].

Putting them together, the Ore polynomial anncii[0] is indeed applied to $\Sigma_{i,n}^{(1)}$; the annihilation is illustrated numerically below.

```
In[ ]:= ClearAll[partunderannc0];
MAX = 15;
For[bb = 0, bb ≤ BB, bb++,
  partunderannc0[bb] = ApplyOreOperator[anncii[0][[1]], part[bb]];
 ];
Table[Sum[partunderannc0[bb] /. {n → nn, ff → matc}, {bb, 0, BB}], {nn, ii + 2, ii + MAX}]

Out[ ]= {0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0}
```

Note that the b-th part acted by anncii[0] can be rewritten as the univariate sequence $c_{n,b}$ acted by a certain Ore polynomial O_b in the algebra $S[n]$. Now we derive an annihilating ideal for each $O_b \cdot c_{n,b}$.

```
In[ ]:= ClearAll[orepolypart, annpart];
For[bb = 0, bb ≤ BB, bb++,
  orepolypart[bb] = ToOrePolynomial[partunderannc0[bb], ff[n, bb]];
  annpart[bb] = DFiniteOreAction[anncii[bb], orepolypart[bb]];
 ];
```

We can indeed check the correctness of these annihilating ideals numerically.

```
In[ ]:= ClearAll[gg];
ClearAll[recpart, valpart];
MAX = 15;
For[bb = 0, bb ≤ BB, bb++,
  recpart[bb] = ApplyOreOperator[annpart[bb], gg[n]][[1]];
  valpart[bb][nn_] := partunderannc0[bb] /. {n → nn, ff → matc};
  Print["Part ", bb, ":\n", Table[
    recpart[bb] /. {n → nn, gg → valpart[bb]}, {nn, ii + 2, ii + MAX}], "\n=========="];
 ];

Part 0:
{0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0}
==========

Part 1:
{0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0}
==========

Part 2:
{0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0}
==========
```

By the closure property, we obtain an annihilating ideal for $\sum_b O_b \cdot c_{n,b} = \text{anncii}[0] \cdot \Sigma_{i,n}^{(1)}$. We then compute the singularities.

```
In[ ]:= anntotal = DFinitePlus[Sequence @@ Table[annpart[bb], {bb, 0, BB}]];

In[ ]:= LeadingExponent[anntotal[[1]]][[1]]

Out[ ]= 13

In[ ]:= Solve[
  (LeadingCoefficient[anntotal[[1]]] /. {n → n - LeadingExponent[anntotal[[1]]][[1]]}) ==
   0, n, Integers]

Out[ ]= {{n → -7}, {n → -7}, {n → -6}, {n → -6}, {n → -3}, {n → -2}, {n → -1}, {n → 0}}
```

We check that the initial values for $\sum_b O_b \cdot c_{n,b} = \text{anncii}[0] \cdot \Sigma_{i,n}^{(1)}$ are all 0, and hence they vanish for all

$n \geq i + 2$. **This confirms that $\Sigma^{(1)}_{i,n}$ is annihilated by anncii[0].**

In[∘]:=
```
Table[Sum[partunderannc0[bb] /. {n → nn, ff → matc}, {bb, 0, BB}],
  {nn, ii + 2, ii + 2 + LeadingExponent[anntotal[[1]]][[1]]}]

ClearAll[sigma1, hh];
sigma[nn_] := Sum[mata1[ii, j] * matc[nn, j], {j, 0, nn - 1}];
sigma1underannc0 = ApplyOreOperator[anncii[0][[1]], hh[n]];
Table[sigma1underannc0 /. {n → nn, hh → sigma},
  {nn, ii + 2, ii + 2 + LeadingExponent[anntotal[[1]]][[1]]}]
```

Out[∘]= {0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0}

Out[∘]= {0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0}

Note that $a2_{i,j}$ equals Binomial[$-i$, 2] when $j = 0$, and 0 when $j \geq 1$ for our choice of i. **Hence, $\Sigma^{(2)}_{i,n} = \sum_{j=0}^{n-1} a2_{i,j}\, c_{n,j}$ = Binomial[$-i$, 2] $* c_{n,0}$ is also annihilated by anncii[0].**

Consequently, $\Sigma_{i,n}$ is annihilated by anncii[0].

Finally, after computing the singularities, we check that $\Sigma_{i,n}$ vanishes for its initial values and hence for all $n \geq i + 2$.

In[∘]:=
```
LeadingExponent[anncii[0][[1]]][[1]]
```

Out[∘]= 5

In[∘]:=
```
Solve[(LeadingCoefficient[anncii[0][[1]]] /.
     {n → n - LeadingExponent[anncii[0][[1]]][[1]]}) == 0, n, Integers]
```

Out[∘]= {{n → 0}, {n → 1}}

In[∘]:=
```
Table[Sum[mata[ii, j] * matc[n, j], {j, 0, n - 1}],
  {n, ii + 2, ii + 2 + LeadingExponent[anncii[0][[1]]][[1]]}]
```

Out[∘]= {0, 0, 0, 0, 0, 0}

Check values at $i = 4$.

In[∘]:=
```
ii = 4;
```

Apply creative telescoping to $a1_{i,j} * c_{n,j}$.

In[∘]:=
```
annii = DFiniteTimesHyper[annc, mata1[ii, j]];
{anniiCT, deltaiiCT} = FindCreativeTelescoping[annii, S[j] - 1];
```

The telescoper part is 1. We then find that the inhomogeneous part is **nonvanishing**, and hence the analysis becomes trickier.

In[∘]:=
```
anniiCT
```

Out[∘]= {1}

Here we note that the delta part is supported on a collection of power products $S[n]^a\, S[j]^b$ with, in particular, $0 \leq b \leq 2$.

In[∘]:=
```
Support[deltaiiCT[[1]]]
```

Out[∘]= $\{\{S_n S_j, S_j^2, S_n, S_j, 1\}\}$

In[∘]:=
```
deltaiiCT[[1, 1]][[1, All, 2]]
BB = Max[%[[All, 2]]]
```

Out[∘]= {{1, 1}, {0, 2}, {1, 0}, {0, 1}, {0, 0}}

Out[∘]= 2

We write explicitly the inhomogeneous part by calling $c_{n,j}$ temporarily under the name ff[n,j].

```
In[∘]:= ClearAll[ff];
inhomii = ApplyOreOperator[deltaiiCT[[1]], mata1[ii, j] * ff[n, j]][[1]];
```

The inhomogeneous part at $j = n$ vanishes by recalling that $c_{n,n-1} = 1$ and $c_{n,j} = 0$ when $j \geq n$.

```
In[∘]:= sumiin = (inhomii /. {j → n}) /.
   {(ff[a_, b_] /; (SameQ[a - b, 1])) → 1, (ff[a_, b_] /; (a - b ≤ 0)) → 0}
```

Out[∘]= 0

The inhomogeneous part at $j = 0$ will be split into parts by collecting $c_{?,b}$.

```
In[∘]:= sumii = inhomii /. {j → 0};
```

```
In[∘]:= ClearAll[part];
For[bb = 0, bb ≤ BB, bb++,
   part[bb] = sumii /. {(ff[a_, b_] /; SameQ[b, bb] == False) → 0};
  ];
```

We shall see that by combining these parts, our $\Sigma^{(1)}_{i,n}$ will be recovered.

```
In[∘]:= MAX = 3;
For[bb = 0, bb ≤ BB, bb++,
   Print["Part ", bb, ":\n",
     Table[part[bb] /. {n → nn, ff → matc}, {nn, ii + 2, ii + 2 + MAX}], "\n=========="];
  ];
Print["Total:\n", Table[Sum[part[bb] /. {n → nn, ff → matc}, {bb, 0, BB}],
    {nn, ii + 2, ii + 2 + MAX}], "\n====================\n",
   "Compare with ∑_ij", Subscript["a1", ToString[ii] <> ",j"], "c_n,j:\n",
   Table[Sum[mata1[ii, j] * matc[nn, j], {j, 0, nn - 1}], {nn, ii + 2, ii + 2 + MAX}]];
```

Part 0:

$\left\{\frac{762186052}{116883}, -\frac{54891850000707}{22355536385}, \frac{3668870038780}{2829090411}, -\frac{3763656233775616}{4848089848875}\right\}$

==========

Part 1:

$\left\{-\frac{132140478871}{20139840}, \frac{54577223610307}{22355536385}, -\frac{3626168218108}{2829090411}, \frac{193954214577664}{255162623625}\right\}$

==========

Part 2:

$\left\{\frac{377477519}{7076160}, 0, 0, 0\right\}$

==========

Total:

$\left\{\frac{512}{39}, -\frac{164480}{11687}, \frac{882944}{58497}, -\frac{43417600}{2680539}\right\}$

====================

Compare with $\sum_{j} a1_{4,j} c_{n,j}$:

$\left\{\frac{512}{39}, -\frac{164480}{11687}, \frac{882944}{58497}, -\frac{43417600}{2680539}\right\}$

In[∘]:=
```
MAX = 10;
Table[Sum[mata1[ii, j] * matc[nn, j], {j, 0, nn - 1}], {nn, ii + 2, ii + MAX}] ==
 Table[Sum[part[bb] /. {n → nn, ff → matc}, {bb, 0, BB}], {nn, ii + 2, ii + MAX}]
```
Out[∘]=
```
True
```

Derive separately the annihilating ideal anncii[b] for the univariate sequence $c_{n,j}$ with fixed $j = b$.

In[∘]:=
```
ClearAll[anncii];
For[bb = 0, bb ≤ BB, bb++,
  anncii[bb] = DFiniteSubstitute[annc, {j → bb}];
 ];
```

In what follows, we show that $\Sigma_{i,n}^{(1)}$ is annihilated by the annihilating ideal anncii[0] for $c_{n,0}$.

For each part obtained earlier, we apply the Ore polynomial anncii[0].

Putting them together, the Ore polynomial anncii[0] is indeed applied to $\Sigma_{i,n}^{(1)}$; the annihilation is illustrated numerically below.

In[∘]:=
```
ClearAll[partunderannc0];
MAX = 15;
For[bb = 0, bb ≤ BB, bb++,
  partunderannc0[bb] = ApplyOreOperator[anncii[0][[1]], part[bb]];
 ];
Table[Sum[partunderannc0[bb] /. {n → nn, ff → matc}, {bb, 0, BB}], {nn, ii + 2, ii + MAX}]
```
Out[∘]=
```
{0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0}
```

Note that the b-th part acted by anncii[0] can be rewritten as the univariate sequence $c_{n,b}$ acted by a certain Ore polynomial O_b in the algebra $S[n]$. Now we derive an annihilating ideal for each $O_b \cdot c_{n,b}$.

In[∘]:=
```
ClearAll[orepolypart, annpart];
For[bb = 0, bb ≤ BB, bb++,
  orepolypart[bb] = ToOrePolynomial[partunderannc0[bb], ff[n, bb]];
  annpart[bb] = DFiniteOreAction[anncii[bb], orepolypart[bb]];
 ];
```

We can indeed check the correctness of these annihilating ideals numerically.

In[∘]:=
```
ClearAll[gg];
ClearAll[recpart, valpart];
MAX = 15;
For[bb = 0, bb ≤ BB, bb++,
  recpart[bb] = ApplyOreOperator[annpart[bb], gg[n]][[1]];
  valpart[bb][nn_] := partunderannc0[bb] /. {n → nn, ff → matc};
  Print["Part ", bb, ":\n", Table[
    recpart[bb] /. {n → nn, gg → valpart[bb]}, {nn, ii + 2, ii + MAX}], "\n=========="];
 ];
```
```
Part 0:
{0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0}
==========

Part 1:
{0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0}
==========
```

```
Part 2:
{0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0}
==========
```

By the closure property, we obtain an annihilating ideal for $\sum_b O_b \cdot c_{n,b}$ = anncii[0]$\cdot \Sigma_{i,n}^{(1)}$. We then compute the singularities.

```
In[∘]:= anntotal = DFinitePlus[Sequence @@ Table[annpart[bb], {bb, 0, BB}]];

In[∘]:= LeadingExponent[anntotal[[1]]][[1]]

Out[∘]= 13

In[∘]:= Solve[
         (LeadingCoefficient[anntotal[[1]]] /. {n → n - LeadingExponent[anntotal[[1]]][[1]]}) ==
          0, n, Integers]

Out[∘]= {{n → -7}, {n → -7}, {n → -6}, {n → -6}, {n → -3}, {n → -2}, {n → -1}, {n → 0}, {n → 1}}
```

We check that the initial values for $\sum_b O_b \cdot c_{n,b}$ = anncii[0]$\cdot \Sigma_{i,n}^{(1)}$ are all 0, and hence they vanish for all $n \geq i + 2$. **This confirms that $\Sigma_{i,n}^{(1)}$ is annihilated by anncii[0].**

```
In[∘]:= Table[Sum[partunderannc0[bb] /. {n → nn, ff → matc}, {bb, 0, BB}],
         {nn, ii + 2, ii + 2 + LeadingExponent[anntotal[[1]]][[1]]}]

        ClearAll[sigma1, hh];
        sigma[nn_] := Sum[mata1[ii, j] * matc[nn, j], {j, 0, nn - 1}];
        sigma1underannc0 = ApplyOreOperator[anncii[0][[1]], hh[n]];
        Table[sigma1underannc0 /. {n → nn, hh → sigma},
         {nn, ii + 2, ii + 2 + LeadingExponent[anntotal[[1]]][[1]]}]

Out[∘]= {0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0}

Out[∘]= {0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0}
```

Note that $a2_{i,j}$ equals Binomial[$-i$, 2] when $j = 0$, and 0 when $j \geq 1$ for our choice of i. **Hence, $\Sigma_{i,n}^{(2)} = \sum_{j=0}^{n-1} a2_{i,j}\, c_{n,j}$ = Binomial[$-i$, 2]$*c_{n,0}$ is also annihilated by anncii[0].**

Consequently, $\Sigma_{i,n}$ is annihilated by anncii[0].

Finally, after computing the singularities, we check that $\Sigma_{i,n}$ vanishes for its initial values and hence for all $n \geq i + 2$.

```
In[∘]:= LeadingExponent[anncii[0][[1]]][[1]]

Out[∘]= 5

In[∘]:= Solve[(LeadingCoefficient[anncii[0][[1]]] /.
            {n → n - LeadingExponent[anncii[0][[1]]][[1]]}) == 0, n, Integers]

Out[∘]= {{n → 0}, {n → 1}}

In[∘]:= Table[Sum[mata[ii, j] * matc[n, j], {j, 0, n - 1}],
         {n, ii + 2, ii + 2 + LeadingExponent[anncii[0][[1]]][[1]]}]

Out[∘]= {0, 0, 0, 0, 0, 0}
```

Proof of (H3)

Annihilator for $a_{n-1,j} * c_{n,j}$. Recall that $a_{n-1,j}$ is split into two parts $a1_{n-1,j} + a2_{n-1,j}$. **No need to exe-**

cute the following codes again. Instead, import the data directly.

In[∘]:=
```
start = CurrentDate[];

ClearAll[n, j];

annH3Smnd1 = DFiniteTimesHyper[annc, mata1[n - 1, j]];
annH3Smnd2 = DFiniteTimesHyper[annc, mata2[n - 1, j]];
Export["annH3Smnd1.txt", {annH3Smnd1}]
Export["annH3Smnd2.txt", {annH3Smnd2}]

Print["Time used: ", CurrentDate[] - start];
```

Out[∘]= annH3Smnd1.txt

Out[∘]= annH3Smnd2.txt

Time used: 3.37915 s

$a1_{n-1,j} * c_{n,j}$

Import the annihilator for $a1_{n-1,j} * c_{n,j}$.

In[∘]:=
```
annH3Smnd1 = ToExpression[Import["annH3Smnd1.txt"]];
AnnInfo[annH3Smnd1]
```

ByteCount: 812 616

Support: $\{\{S_n^2, S_n S_j, S_j^2, S_n, S_j, 1\}, \{S_j^3, S_n S_j, S_j^2, S_n, S_j, 1\}, \{S_n S_j^2, S_n S_j, S_j^2, S_n, S_j, 1\}\}$

degree {n, j}: {{24, 17}, {21, 20}, {16, 11}}

Standard Monomials: $\{1, S_j, S_n, S_j^2, S_n S_j\}$

Holonomic Rank: 5

Import the telescoper for $a1_{n-1,j} * c_{n,j}$.

In[∘]:=
```
annH3CT1 = ToExpression[Import["annH3CT1.txt"]];
```

In[∘]:=
```
AnnInfo[annH3CT1]
```

ByteCount: 6 102 528

Support:
$\{\{S_n^{20}, S_n^{19}, S_n^{18}, S_n^{17}, S_n^{16}, S_n^{15}, S_n^{14}, S_n^{13}, S_n^{12}, S_n^{11}, S_n^{10}, S_n^{9}, S_n^{8}, S_n^{7}, S_n^{6}, S_n^{5}, S_n^{4}, S_n^{3}, S_n^{2}, S_n, 1\}\}$

degree {n}: {{556}}

Standard Monomials:
$\{1, S_n, S_n^2, S_n^3, S_n^4, S_n^5, S_n^6, S_n^7, S_n^8, S_n^9, S_n^{10}, S_n^{11}, S_n^{12}, S_n^{13}, S_n^{14}, S_n^{15}, S_n^{16}, S_n^{17}, S_n^{18}, S_n^{19}\}$

Holonomic Rank: 20

In[∘]:=
```
deltaH3CT1 = ToExpression[Import["deltaH3CT1.txt"]];
```

In[∘]:=
```
ByteCount[deltaH3CT1]
```

Out[∘]= 3 182 669 576

Longest digits of the coefficients in the annihilator.

In[∘]:=
```
Table[Max[IntegerLength[CoefficientList[annH3CT1[[1]][[1]][[jj]][[1]], n]]],
  {jj, 1, LeadingExponent[annH3CT1[[1]]][[1]]}]
Max[
  %]
```

Out[∘]= {969, 971, 971, 972, 972, 974, 975, 976, 976,
 977, 977, 977, 977, 977, 977, 977, 977, 976, 976, 975}

Out[∘]= 977

Verify the telescopers for $a1_{n-1,j} * c_{n,j}$. **Note that this step is VERY Memory-consuming, so we executed the command on an Amazon AWS cluster.** To illustrate the process, we may nonrigorously substitute the parameters with specific values.

```
In[∘]:= Timing[
         OreReduce[MapThread[(#1 + (S[j] - 1) ** #2) &, {annH3CT1, deltaH3CT1}], annH3Smnd1]]
Out[∘]= $Aborted

In[∘]:= subs = {n → 10};
        {annH3CT1subs, deltaH3CT1subs} =
          OrePolynomialSubstitute[#, subs] & /@ {annH3CT1, deltaH3CT1};
        Timing[OreReduce[MapThread[(#1 + (S[j] - 1) ** #2) &, {annH3CT1subs, deltaH3CT1subs}],
          annH3Smnd1, OrePolynomialSubstitute → subs]]
Out[∘]= {28.9844, {0}}

In[∘]:= subs = {n → 357};
        {annH3CT1subs, deltaH3CT1subs} =
          OrePolynomialSubstitute[#, subs] & /@ {annH3CT1, deltaH3CT1};
        Timing[OreReduce[MapThread[(#1 + (S[j] - 1) ** #2) &, {annH3CT1subs, deltaH3CT1subs}],
          annH3Smnd1, OrePolynomialSubstitute → subs]]
Out[∘]= {39.2813, {0}}
```

$a2_{n-1,j} * c_{n,j}$

Import the annihilator for $a2_{n-1,j} * c_{n,j}$.

```
In[∘]:= annH3Smnd2 = ToExpression[Import["annH3Smnd2.txt"]];
        AnnInfo[annH3Smnd2]
```

ByteCount: 694 568
Support: $\{\{S_n^2, S_n S_j, S_j^2, S_n, S_j, 1\}, \{S_j^3, S_n S_j, S_j^2, S_n, S_j, 1\}, \{S_n S_j^2, S_n S_j, S_j^2, S_n, S_j, 1\}\}$
degree {n, j}: {{23, 14}, {19, 17}, {16, 11}}
Standard Monomials: $\{1, S_j, S_n, S_j^2, S_n S_j\}$
Holonomic Rank: 5

Import the telescoper for $a2_{n-1,j} * c_{n,j}$.

```
In[∘]:= annH3CT2 = ToExpression[Import["annH3CT2.txt"]];

In[∘]:= AnnInfo[annH3CT2]
```

ByteCount: 5 638 048
Support:
$\{\{S_n^{20}, S_n^{19}, S_n^{18}, S_n^{17}, S_n^{16}, S_n^{15}, S_n^{14}, S_n^{13}, S_n^{12}, S_n^{11}, S_n^{10}, S_n^9, S_n^8, S_n^7, S_n^6, S_n^5, S_n^4, S_n^3, S_n^2, S_n, 1\}\}$
degree {n}: {{523}}
Standard Monomials:
$\{1, S_n, S_n^2, S_n^3, S_n^4, S_n^5, S_n^6, S_n^7, S_n^8, S_n^9, S_n^{10}, S_n^{11}, S_n^{12}, S_n^{13}, S_n^{14}, S_n^{15}, S_n^{16}, S_n^{17}, S_n^{18}, S_n^{19}\}$
Holonomic Rank: 20

```
In[∘]:= deltaH3CT2 = ToExpression[Import["deltaH3CT2.txt"]];

In[∘]:= ByteCount[deltaH3CT2]
Out[∘]= 4 665 533 376
```

Longest digits of the coefficients in the annihilator.

In[∘]:=
```
Table[Max[IntegerLength[CoefficientList[annH3CT2[[1]][[1]][[jj]][[1]], n]]],
 {jj, 1, LeadingExponent[annH3CT2[[1]]][[1]]}]
Max[
 %]
```

Out[∘]= {939, 940, 941, 942, 942, 944, 945, 946, 946, 947, 947, 947, 947, 947, 947, 947, 947, 946, 946, 945}

Out[∘]= 947

Verify the telescopers for $a2_{n-1,j} * c_{n,j}$. **Note that this step is VERY Memory-consuming, so we executed the command on an Amazon AWS cluster.** To illustrate the process, we may nonrigorously substitute the parameters with specific values.

In[∘]:=
```
Timing[
 OreReduce[MapThread[(#1 + (S[j] - 1) ** #2) &, {annH3CT2, deltaH3CT2}], annH3Smnd2]]
```

Out[∘]= $Aborted

In[∘]:=
```
subs = {n → 10};
{annH3CT2subs, deltaH3CT2subs} =
  OrePolynomialSubstitute[#, subs] & /@ {annH3CT2, deltaH3CT2};
Timing[OreReduce[MapThread[(#1 + (S[j] - 1) ** #2) &, {annH3CT2subs, deltaH3CT2subs}],
  annH3Smnd2, OrePolynomialSubstitute → subs]]
```

Out[∘]= {34.7188, {0}}

In[∘]:=
```
subs = {n → 357};
{annH3CT2subs, deltaH3CT2subs} =
  OrePolynomialSubstitute[#, subs] & /@ {annH3CT2, deltaH3CT2};
Timing[OreReduce[MapThread[(#1 + (S[j] - 1) ** #2) &, {annH3CT2subs, deltaH3CT2subs}],
  annH3Smnd2, OrePolynomialSubstitute → subs]]
```

Out[∘]= {35.2031, {0}}

$a_{n-1,j} * c_{n,j}$

Check that annH3CT1 and annH3CT2 differ by a scalar. So annH3CT1 also annihilates $\sum_j a2_{n-1,j} * c_{n,j}$.

In[∘]:=
```
annH3CT1LeadingCoefficient = LeadingCoefficient[annH3CT1[[1]]];
annH3CT2LeadingCoefficient = LeadingCoefficient[annH3CT2[[1]]];
Scalar = Factor[annH3CT1LeadingCoefficient / annH3CT2LeadingCoefficient]
```

Out[∘]= $-n\,(1+n)\,(2+n)^2\,(3+n)^2\,(4+n)^2\,(5+n)^2\,(6+n)^2\,(7+n)^2\,(8+n)^2\,(9+n)^2\,(10+n)^2\,(11+n)^2\,(12+n)^2\,(13+n)^2\,(14+n)^2\,(15+n)\,(16+n)\,(17+n)\,(18+n)\,(19+n)$

In[∘]:=
```
annH3CT1CoeffList = OrePolynomialListCoefficients[annH3CT1[[1]]];
annH3CT2CoeffList = OrePolynomialListCoefficients[annH3CT2[[1]]];
```

In[∘]:=
```
Expand[annH3CT1CoeffList] == Expand[Scalar * annH3CT2CoeffList]
```

Out[∘]= True

In[∘]:=
```
annH3CT = annH3CT1;
```

Check initial values.

Check the integer roots of the leading coefficient.

Order of the recurrence for $\sum_j a_{n-1,j} * c_{n,j}$.

In[∘]:=
```
LeadingExponent[annH3CT[[1]]][[1]]
```

Out[∘]= 20

```
In[∘]:= Solve[
  (LeadingCoefficient[annH3CT[[1]]] /. {n → n - LeadingExponent[annH3CT[[1]]][[1]]}) ==
   0, n, Integers]
```

```
Out[∘]= {{n → 1}, {n → 1}, {n → 2}, {n → 2}, {n → 3}, {n → 3}, {n → 4}, {n → 4},
  {n → 5}, {n → 5}, {n → 6}, {n → 6}, {n → 7}, {n → 7}, {n → 8}, {n → 8},
  {n → 9}, {n → 9}, {n → 10}, {n → 10}, {n → 11}, {n → 11}, {n → 12}, {n → 12},
  {n → 13}, {n → 13}, {n → 14}, {n → 14}, {n → 15}, {n → 15}, {n → 16}, {n → 16},
  {n → 17}, {n → 17}, {n → 18}, {n → 18}, {n → 19}, {n → 19}, {n → 20}, {n → 20}}
```

Simplify the quotient prodform[n]/prodform[n-1].

```
In[∘]:= quot = prodform[n] / prodform[n - 1] /.
      prod[f_, {i, a_, n}] :> (f /. i → n) * prod[f, {i, a, n - 1}];
   quot = FunctionExpand[quot /. prodsimp /. prod → Product]
```

Out[∘]= $\dfrac{\text{Gamma}\left[\frac{1}{2}+\frac{n}{4}\right]\,\text{Gamma}[-1+6\,n]}{2\,(-1+2\,n)\,\text{Gamma}\left[-\frac{1}{2}+\frac{5\,n}{4}\right]\,\text{Gamma}[-1+5\,n]}$

Verify the quotient prodform[n]/prodform[n-1] also satisfies the recurrence for $\sum_j a_{n-1,j} * c_{n,j}$.

```
In[∘]:= Timing[OreReduce[annH3CT[[1]], Annihilator[quot, S[n]]]]
```

Out[∘]= {1.79688, 0}

Check the first few (more than necessary) initial values.

```
In[∘]:= Table[Sum[mata[n - 1, j] * matc[n, j], {j, 0, n - 1}] == prodform[n] / prodform[n - 1],
  {n, 1, LeadingExponent[annH3CT[[1]]][[1]] + 10}]
```

```
Out[∘]= {True, True, True, True, True, True, True, True, True, True, True, True, True, True, True,
  True, True, True, True, True, True, True, True, True, True, True, True, True, True, True}
```